
HP-35S Calculator Program Closure 1A

Traverse Closure with Area Calculation and Co-ordinates

Programmer: Dr. Bill Hazelton

Date: October, 2007.

Line Instruction Display User Instructions
A001 LBL A Press  XEQ  A  ENTER
A002 CLSTK
A003 SF  10
A004 USE COORDS USE COORDS (Key in using EQN RCL U, RCL S,  etc.)
A005 PSE
A006 INPUT  C C?
A007 RCL  C
A008 x = 0?
A009 GTO  A021
A010 ENTER N0 ENTER N0 (Key in using EQN RCL E, RCL N, etc.)
A011 PSE
A012 INPUT  N N?
A013 ENTER E0 ENTER E0 (Key in using EQN RCL E, RCL N, etc.)
A014 PSE
A015 INPUT  E E?
A016 RCL  E
A017 0 i 1 (Key in as 0, then i, then 1, press ENTER.)
A018 ×
A019 RCL+  N
A020 STO  F
A021 360
A022 STO  B
A023 CLSTK
A024 STO  P
A025 STO  Q
A026 STO  M
A027 XEQ  V001
A028 STO  R
A029 STO+  P
A030 ARG
A031 RCL  P
A032 ARG
A033 –
A034 SIN
A035 STO  S
A036 RCL  R
A037 ABS
A038 STO×  S
A039 RCL  P

A040 ABS
A041 STO×  S
A042 RCL  S
A043 2
A044 ÷
A045 STO+  Q
A046 1
A047 STO+  M
A048 RCL  C
A049 x = 0?
A050 GTO  A068
A051 RCL  P
A052 RCL+  F
A053 XEQ  X001
A054 RCL  M
A055 RCL  Q
A056 ABS
A057 R↓
A058 STOP
A059 RCL  P
A060 ARG
A061 x < 0?
A062 RCL+  B
A063 →HMS
A064 RCL  P
A065 ABS
A066 STOP
A067 GTO  A027
A068 RCL  Q
A069 ABS
A070 RCL  P
A071 ARG
A072 x < 0?
A073 RCL+  B
A074 →HMS
A075 RCL  P
A076 ABS
A077 RCL  M
A078 STOP
A079 GTO  A027

This version of the program is designed to completely hide the complex number work that the
calculator performs to compute the traverse. All values entered and returned are presented in
much the same manner as with the HP-33S calculator.

Notes

(1) Set the calculator into DEGREES mode (press MODE 1) before starting.

(2) This is a general traverse closure program that computes the azimuth and
distance, plus area, to each point around the traverse, together with co-ordinates,
if desired. It also computes the misclosure of a closed traverse.

(3) This program uses the V program as a sub-routine for data entry, so Program V
must be in the calculator as program V (or the XEQ V001 at line A027 changed
to reflect the changed label. The V program allows entry of azimuths in
D.MMSS format (HP notation) and distances, while converting them to the
internal format. Program V is one of the HP-35s Utilities programs (Utility 3).

(4) This program uses the X program as a sub-routine for co-ordinate return from the
internal format., so program X must be in the calculator as program X (or the
XEQ X001 at line A053 changed to reflect the label change). The X program
allows co-ordinates to be returned from the internal processing format of the
calculator. Program X is one of the HP-35s Utilities programs (Utility 4).

(5) The program allows the user to choose if the co-ordinates of each point are to be
calculated. The user is prompted with USE COORDS briefly, followed by the C?
prompt. If co-ordinates are desired, key in 1, if not, key in 0, then press R/S to
continue.

(6) After each side (azimuth and distance) has been entered, the calculator produces
the following output.

    A. If the user has selected to use co-ordinates, the calculator has the following data
on the stack. It will stop and display this information, with the number of sides
entered in the display in line 2, and the Easting co-ordinate of the point in line 1.

Stack Register Contents

T Area of the traverse thus far

Z Northing co-ordinate of current point

Y Easting co-ordinate of current point

X Number of sides entered

The user can scroll through the stack, using the R↓ key, and can perform any
other operation of interest to the data on the stack. This information is stored in
memory registers for use later in the program, so the stack may be changed and
worked with as needed. The user can also continue without viewing the stack.

When the user presses R/S, the calculator takes the line to the current forward
point from the starting point, and converts it into the distance (which is placed in
line 2, the X register) and the azimuth in degrees, minutes and seconds (HP
notation) in line 1 of the display (the Y register).

Stack Register Contents

T

Z

Y Azimuth from start to current point (in D.MMSSss)

X Distance from start to current point

When the user presses R/S again, the calculator prompts for the azimuth of the
next side to be entered. The azimuth should be entered in HP notation
(DDD.MMSSss).

    B. If the user has selected not to show co-ordinates, the calculator has the following
data on the stack. It will stop and display this information, with the number of
sides entered in line 2 of the display (the X register) and the distance of the
misclosure or the line connecting the starting point to the current point, in line 1
of the display (Y register).

Stack Register Contents

T Area of the traverse thus far

Z Azimuth of the line from the start to current point (in HP notation)

Y Distance of the line from the start to the current point

X Number of sides entered

By pressing the R↓ key, the user can see the azimuth in register Z of the stack.
Pressing R↓ again will show the area to the current point.

If the rectangular components of the misclosure (or the line from the start to the
current forward point) are needed, press RCL  P, then XEQ  X  ENTER. The
Easting component of the misclosure (∆E) will be displayed in the X register
(line 2), while the Northing component (∆N) will be displayed in the Y register
(line 1).

When the user presses R/S again, the calculator prompts for the azimuth of the
next side.

(7) Azimuths are entered and displayed by themselves in HP notation, i.e.,
DDD.MMSSss.

(8) This program forms the basis of the two missing distances (2MD) program. Enter
the known sides using this program to begin the 2MD computation process.

(9) In order to display the prompts, this program sets Flag 10. However, the program
never ends, because it is up to the user to decide when to stop and move control
elsewhere. So the program never clears Flag 10. If you require Flag 10 to be
clear, in order to process equations, you must do this manually.



HP-35S Calculator Program Closure   1A
Traverse Closure with Area Calculation and Co-ordinates

- 5-
 -

Theory

The traverse closure programs works by converting the entered azimuths (in DDD.MMSS, or HP,
notation) and distances into complex numbers (which act as 2-D vectors), which are then added to
compute the location of points around the traverse. The area is computed by triangles developed
by each new side of the traverse and the line from the starting point to the current forward point,
and is updated with each new side. So the area is that of the polygon formed by the traverse
entered thus far and the line from the start to the current point. This allows areas to be
incremented for lot splitting calculations.

The azimuth and distance of the line from the start to the current point is also placed on the stack
after each line. This allows a connecting line to be computed easily between two points. The final
azimuth and distance is the traverse misclosure and the area is that of the traverse.

If the user chooses, the co-ordinates of the starting point may be entered, and if this choice is
made, the calculator displays the co-ordinates of each point, in addition to the other information.

An arbitrary azimuth is satisfactory. Plane surveying assumptions apply. The program uses no
error checking on entered data.

Sample Computation

Bearing Distance

6° 53' 10" 72.00

112° 37' 20" 102.23

185° 39' 50" 29.04

181° 30' 00" 27.88

283° 54' 30" 102.38

Final Results DE = 0.0228

DN = –0.0022

Misclosure Length = 0.0229

Misclosure Bearing =    95° 24' 15"

Area =    6,378.4660

Stepping through the Calculation

A. Without Co-ordinates

Press  XEQ  A  ENTER

Calculator prompts with USE COORDS, the C?

Key in 0, then press R/S.

Side 1

Calculator prompts with A? for azimuth of side.

Key in 6.5310, press R/S.

Calculator prompts with D? for distance of side.

Key in 72.00, press R/S.

Display shows: 72.0000 (distance from start)
1.0000 (number of sides entered)

Press the R↓ key twice, and the display becomes:

0.0000 (area thus far)
6.5310 (azimuth from start in HP notation (D.MMSS))

Press R/S.

Side 2

Calculator prompts with A? for azimuth of side.

Key in 112.372, press R/S.

Calculator prompts with D? for distance of side.

Key in 102.23, press R/S.

Display shows: 107.9004 (distance from start)
2.0000 (number of sides entered)

Press the R↓ key twice, and the display becomes:

3,542.3468 (area thus far)
72.3939 (azimuth from start in HP notation (D.MMSS))

Press R/S.

Side 3

Calculator prompts with A? for azimuth of side.

Key in 185.395, press R/S.

Calculator prompts with D? for distance of side.

Key in 29.04, press R/S.

Display shows: 100.1841 (distance from start)
3.0000 (number of sides entered)

Press the R↓ key twice, and the display becomes:

4,984.4807 (area thus far)
88.0808 (azimuth from start in HP notation (D.MMSS))

Press R/S.

Side 4

Calculator prompts with A? for azimuth of side.

Key in 181.3, press R/S.

Calculator prompts with D? for distance of side.

Key in 27.88, press R/S.

Display shows: 102.4027 (distance from start)
4.0000 (number of sides entered)

Press the R↓ key twice, and the display becomes:

6378.6396 (area thus far)
103.5423 (azimuth from start in HP notation (D.MMSS))

Press R/S.

Side 5

Calculator prompts with A? for azimuth of side.

Key in 283.543, press R/S.

Calculator prompts with D? for distance of side.

Key in 102.38, press R/S.

Display shows: 0.0229 (distance from start, also linear misclosure)
5.0000 (number of sides entered)

Press the R↓ key twice, and the display becomes:

6378.4660 (area thus far)
95.2415 (azimuth from start, also azimuth of misclosure)

Pressing  RCL  P, then XEQ  X  ENTER  brings the misclosure to the display in rectangular form.
The Northing component of the misclosure (∆N) is in the Y register (line 1) and is –0.0022, while
the Easting component of the misclosure (∆E) is in the X register (line 2) and  is 0.0228.

B. Using Co-ordinates

Press  XEQ  A  ENTER

Calculator prompts with USE COORDS briefly, then C?

Key in 1, then press R/S.

Calculator prompts ENTER N0 briefly, then N?

Key in 1000.000, press R/S.

Calculator prompts ENTER E0 briefly, then E?

Key in 500.000, press R/S.

Side 1

Calculator prompts with A? for azimuth of side.

Key in 6.5310, press R/S.

Calculator prompts with D? for distance of side.

Key in 72.00, press R/S.

Display shows: 508.6325 (Easting co-ordinate of current forward point)
1.0000 (number of sides entered)

Press the R↓ key twice, and the display becomes:

0.0000 (area thus far)
1,071.4806 (Northing co-ordinate of current forward point)

Press R/S.

Display shows: 6.5310 (Azimuth for current forward point from starting point)
72.0000 (Distance from starting point to current forward point)

Press R/S.

Side 2

Calculator prompts with A? for azimuth of side.

Key in 112.372, press R/S.

Calculator prompts with D? for distance of side.

Key in 102.23, press R/S.

Display shows: 602.9971 (Easting co-ordinate of current forward point)
2.0000 (number of sides entered)

Press the R↓ key twice, and the display becomes:

3,542.3468 (area thus far)
1,032.1575 (Northing co-ordinate of current forward point)

Press R/S.

Display shows: 72.3939 (Azimuth for current forward point from starting point)
107.9004 (Distance from starting point to current forward point)

Press R/S.

Side 3

Calculator prompts with A? for azimuth of side.

Key in 185.395, press R/S.

Calculator prompts with D? for distance of side.

Key in 29.04, press R/S.

Display shows: 600.1310 (Easting co-ordinate of current forward point)
3.0000 (number of sides entered)

Press the R↓ key twice, and the display becomes:

4,984.4807 (area thus far)
1,003.2593 (Northing co-ordinate of current forward point)

Press R/S.

Display shows: 88.0808 (Azimuth for current forward point from starting point)
100.1841 (Distance from starting point to current forward point)

Press R/S.

Side 4

Calculator prompts with A? for azimuth of side.

Key in 181.3, press R/S.

Calculator prompts with D? for distance of side.

Key in 27.88, press R/S.

Display shows: 599.4012 (Easting co-ordinate of current forward point)
4.0000 (number of sides entered)

Press the R↓ key, and the display becomes:

6,378.6396 (area thus far)
975.3888 (Northing co-ordinate of current forward point)

Press R/S.

Display shows: 103.5423 (Azimuth for current forward point from starting point)
102.4027 (Distance from starting point to current forward point)

Press R/S.

Side 5

Calculator prompts with A? for azimuth of side.

Key in 283.543, press R/S.

Calculator prompts with D? for distance of side.

Key in 102.38, press R/S.

Display shows: 500.0228 (Easting co-ordinate of current forward point)
5.0000 (number of sides entered)

Press the R↓ key, and the display becomes:

6,378.4660 (area thus far)
999.9978 (Northing co-ordinate of current forward point)

Press R/S.

Display shows: 95.2415 (Azimuth for current forward point from starting point)
0.0229 (Distance from starting point to current forward point)

Press  RCL  P, then XEQ  X  ENTER to bring the misclosure onto the stack in rectangular mode.
The misclosure in Northings (latitude, ∆N) will be displayed in the Y register (line 1) as –0.0022.
The misclosure in Eastings (departure, ∆E) will be displayed in the X register (line 2) and is
0.0228.

Storage Registers Used

A Used by the V program for the entered azimuth.

B Stores 360 for azimuth correction.

C Test for displaying co-ordinates: 1 = YES; 0 = NO.

D Used by the V program for the entered distance.

E Easting co-ordinates of the starting point.

F Co-ordinates of starting point, as a complex number.

I Used by the V and X programs to address the additional storage registers.

M The number of sides entered.

N The Northing co-ordinate of the starting point.

P Current position of forward point, as a complex number.

Q Current area.

R Last side entered, as a complex number.

S Temporary storage for area calculation.

Statistical Registers: not used.

Other registers: 10, 11 and 12 used by the V program; 10 used by the X program.

Labels Used

Label A Length  =   271 Checksum  =  1175

Use the length (LN=) and Checksum (CK=) values to check if program was entered correctly.
Use the sample computation to check proper operation after entry.

Routines Called

The program labeled V, which takes an azimuth in degrees, minutes and seconds (in HP
notation), and a distance, and converts them to a complex number for processing in the calculator.
This routines uses storage location A and D, but copies out and replaces the contents of these
storage locations in order to preserve them. It also uses storage register I for indirect addressing,
which uses registers 10, 11 and 12.

The program labeled X, which takes a complex number and converts it to the rectangular
components, is called for co-ordinate presentation in this program. It uses storage register I for
indirect addressing, which uses register 10.

Label V Length  =   128 Checksum  =  39FE

Label X Length  =     53 Checksum  =  C46D


